/*	This program produces Table C3 */

***** Set directories 
local dir_clean 	"~/Dropbox/Retirement gaming/clean"
local dir_output 	"~/Dropbox/Retirement gaming/output/dataverse"


use "`dir_clean'/mainsample_medbcw.dta", clear

* Sets of controls		
local listcontrols1 = " i.t  " 
local listcontrols2 = " i.t i.ciiu2_1stobs i.ndep_cat_1stobs i.ndep_cat " 
local listcontrols3 = " i.t i.ciiu2_1stobs i.ndep_cat_1stobs i.ndep_cat i.year#i.i.ciiu2_1stobs i.year#i.ndep_cat_1stobs "  



**** EMPLOYED ONLY IN SMALL FIRMS, HETEROGENEITY 
*** SEMI PARAMETRIC (Table C.3)
estimates clear
* 70%
foreach c in  1 2 3 {
		preserve
		keep if  empl==1 & prop_mempl>=.7 &  large_1stobs==0  & tenure1yrs_at49==1
		*Normalize wages
		sum W if time_bcw==-1
		local meanw=r(mean)
		g Wz=W/`meanw'
		* Regress
		reghdfe Wz pre6 pre25 post03 post4, absorb(`listcontrols3') 
		g thissample=e(sample)
		eststo: reghdfe Wz pre6 pre25 post03 post4 if thissample==1, absorb(`listcontrols`c'') vce(cluster i )
		restore
}
* MATCHED WORKER-FIRM OBSERVATIONS
foreach c in  1 2 3 {
	*Keep those observed at least X% of the time
		preserve
		keep if  empl==1 & prop_mempl>=.7 &  large_1stobs==0  & match_firm49==1 & tenure1yrs_at49==1
		*Normalize wages
		sum W if time_bcw==-1
		local meanw=r(mean)
		g Wz=W/`meanw'
		* Regress
		reghdfe Wz pre6 pre25 post03 post4, absorb(`listcontrols3') 
		g thissample=e(sample)
		eststo: reghdfe Wz pre6 pre25 post03 post4 if thissample==1, absorb(`listcontrols`c'') vce(cluster i )
		restore
}
* Table with results 
esttab using "`dir_output'/tableC3.tex", ///
	keep(pre25 post03 post4)  stats(N N_clust1, label("Observations" "Individuals") fmt(%9.0gc)) ///
	b(%4.3f) se(%4.3f) star(* 0.10 ** 0.05 *** 0.01) ///
	f nomtitles  nogaps replace label nonumbers nolines	prefoot(\hline)
	
clear all
exit
